



[Startseite](#) > [Foren](#) > [Reporting & Visualisierung](#)

Graphiken mit Vermerk 'n = Anzahl' versehen

5 September, 2011 - 19:09 — felizitas

Hi,

ich wüsste gern, wie ich in Graphiken wie Boxplots, Histo- oder Balkendiagrammen automatisch 'n = Anzahl' ausgeben lassen kann und wie ich Balken in Balkendiagrammen mit der jeweiligen Anzahl beschriften lassen kann, vielen Dank!

Viele Grüße aus Berlin

* felizitas *

Foren:

[Reporting & Visualisierung](#)

[Log in](#) or [register](#) to post comments

N vorher in Macrovariable speichern

6 September, 2011 - 11:34 — JanHeuer

Hallo felizitas,

man kann die Anzahl vorher ermitteln und in einer Macrovariable speichern. Diesen Wert kann man dann im Title- oder Footnote Statement mit ausgeben.

Die Ausgabe von der Anzahl in Balkendiagrammen erfolgt durch Outside in Proc Gchart.

```
/* Anzahl der Datensätze bestimmen und in Macrovariable abspeichern */
Proc Sql noprint;
  Select Count(Age) Into :Anzahl From Sashelp.Class;
Quit;
/* Ausgabe vorbereiten */
Ods Html;
Axis1
  Style = 0
  Width = 1
  Major = None
  Minor = None
  Value = None
  Label = (" ")
;
Axis2
  Style = 1
  Width = 1
  Label = ("Alter")
;
Title h = 3 "Verteilung Alter in Sashelp.Class";
/* Ausgabe von N in die Fußnote */
```

```
Footnote h = 1 "(n = %trim(&Anzahl.))";
Proc Gchart Data = Sashelp.Class;
  Vbar Age /
  Frame Discrete
  Type      = Freq    /* Statistik */
  Outside   = Freq    /* zeigt die Statistik außerhalb des Balkens an */
  Coutline  = Black
  Raxis     = Axis1
  Maxis     = Axis2
;
Run;
/* Aufräumen */
Title;Footnote;
%Symdel Anzahl;
Quit;
Ods Html Close;
```

Schöne Grüße Jan

[Log in](#) or [register](#) to post comments

Bei der PROC GPLOT gibt es

6 September, 2011 - 13:23 — Chris86

Bei der PROC GPLOT gibt es die Möglichkeit auch über

```
SYMBOL<1:99> pointlabel=("#name der variablen");
```

zu benennen. Die angesprochene Variable muss dann nur im dataset, welches gezeichnet wird, vorhanden sein.

Grüße,
Chris86

[Log in](#) or [register](#) to post comments
